

```

1 REM *****
2 REM ***** SOLITARIO DE LOS ESPACIOS *****
3 REM *****-----*****
4 REM ***** ANGEL LARUMBE DORAL *****
5 REM *****
10 RANDOMIZE TIME
20 WINDOW#4,1,80,1,1
30 WINDOW#5,1,1,1,24
40 SYMBOL 250,126,60,153,255,255,219,153,24
50 MODE 2:DIM posic$(4,13),da(13,4):LOCATE 30,13:PRINT "COLOCANDO CARTAS"
55 REM ----- REPARTO DE LAS CARTAS EN SUS POSICIONES -----
60 FOR i=1 TO 13
70 FOR j=1 TO 4
80 c=INT(RND*13)+1:p=INT(RND*4)+1:IF da(c,p)=1 THEN 80
90 REM posic$ almacena las cartas colocadas en esas posiciones.
100 posic$(j,i)=STR$(c)+","+STR$(p):da(c,p)=1:NEXT j,i
110 coli=2:colf=coli+4:filai=2:filaf=filai+3:CLS
120 WINDOW#1,1,80,22,23
130 WINDOW#2,coli,colf,filai,filaf
140 PAPER#2,3:PEN#2,0
150 FOR i=1 TO 4
155 REM ----- IMPRESION POR PANTALLA DE LAS CARTAS RESULTANTES -----
160 FOR j=1 TO 13
170 LOCATE#4,coli+1,1:FRINT#4,USING "##";j
180 WINDOW#2,coli,colf,filai,filaf
190 IF VAL(LEFT$(posic$(i,j),3))=1 THEN 300
195 GOSUB 200:GOTO 290
197 REM ----- SUBROUTINA PARA IMPRIMIR 'J' POR 11,ETC... -----
200 CLS#2:v$=RIGHT$(posic$(i,j),1)
210 IF VAL(v$)=1 THEN palo$=CHR$(227):GOTO 240
220 IF VAL(v$)=2 THEN palo$=CHR$(229):GOTO 240
230 IF VAL(v$)=3 THEN palo$=CHR$(228) ELSE palo$=CHR$(250)
240 carta=VAL(LEFT$(posic$(i,j),3))
250 IF carta=11 THEN pon$="J":GOTO 280
260 IF carta=12 THEN pon$="Q":GOTO 280
270 IF carta=13 THEN pon$="K" ELSE pon$=STR$(carta)
280 PRINT#2,pon$;" ";palo$:RETURN
290 IF j=13 THEN 310
300 coli=coli+6:colf=coli+4
310 NEXT j
320 coli=2:colf=coli+4:filai=filai+5:filaf=filai+3
330 NEXT i
335 REM ----- ELECCION DE LA CARTA A CAMBIAR DE SITIO -----
340 CLS#1:INPUT#1,"Fila de origen ",f$:f=VAL(f$)
350 IF f<1 OR f>4 THEN 340
360 INPUT#1,"Columna de Origen ",c$:coli=2:colf=6:filai=2:filaf=5
370 c=VAL(c$):IF c<1 OR c>13 THEN 360
375 REM ----- BUSQUEDA DE LA POSICION DE LA CARTA ELEGIDA -----
380 FOR i=1 TO f
390 FOR j=1 TO 13
400 IF i=f AND j=c THEN 450
410 coli=coli+6:colf=coli+4
420 NEXT j
430 coli=2:colf=coli+4:filai=filai+5:filaf=filai+3
440 NEXT i
450 WINDOW#2,coli,colf,filai,filaf
460 PAPER#2,2:PEN#2,11
470 carta=VAL(LEFT$(posic$(f,c),3)):IF carta=1 THEN PRINT CHR$(7);:GOTO 340

```

```

472 REM ----- MARCA LA CARTA ELEGIDA CON FONDO OSCURO -----
475 GOSUB 200
500 REM ----- SOLICITA EL DESTINO DE LA CARTA ELEGIDA -----
560 CLS#1:INPUT#1,"Fila destino ",fd$:fd=VAL(fd$)
570 IF fd<1 OR fd>4 THEN 560
580 INPUT#1,"Columna destino ",cd$:cd=VAL(cd$)
590 IF cd<1 OR cd>13 THEN 580
595 REM ----- COMPRUEBA QUE SE PUEDA COLOCAR -----
600 cartad=VAL(LEFT$(posic$(fd,cd),3))
610 IF cartad<>1 THEN PRINT CHR$(7);:w=1:fd=f:cd=c:GOTO 660
620 IF cd=1 AND carta<>2 THEN w=1:PRINT CHR$(7);:fd=f:cd=c:GOTO 660
630 x=cd-1:IF x<1 THEN x=1
640 cartaa=VAL(LEFT$(posic$(fd,x),3))
650 IF (carta-cartaa<>1) THEN fd=f:cd=c:w=1:PRINT CHR$(7);
655 IF (carta-cartaa<>1) AND (carta<>2) THEN PRINT CHR$(7);:fd=f:cd=c:w=1
660 coli=2:colf=6:filai=2:filaf=5
670 IF cartaa=1 AND carta<>2 THEN w=1
675 REM ----- SI W=0 PUEDE COLOCARSE:PONE LA CARTA EN COORD. CALC. -----
680 CLS#2:FOR i=1 TO fd
690 FOR j=1 TO 13
700 IF i=fd AND j=cd THEN 750
710 coli=coli+6:colf=coli+4
720 NEXT j
730 coli=2:colf=coli+4:filai=filai+5:filaf=filai+3
740 NEXT i
750 WINDOW#2,coli,colf,filai,filaf
760 PAPER#2,3:PEN#2,0:CLS#2:PRINT#2,pon$;" ";palo$
765 REM ----- SI W=1 NO PUEDE PONERSE:VUELVE A DIBUJAR LA ELEGIDA -----
770 IF w=1 THEN w=0:GOTO 340
775 REM ---- PUEDE COLOCARSE:INTERCAMBIA VALORES DE LAS POS. -----
780 aux$=posic$(fd,cd)
790 posic$(fd,cd)=posic$(f,c)
800 posic$(f,c)=aux$
805 REM ----- COMPRUEBA SI ES FINAL DE PARTIDA -----
810 FOR i=1 TO 4
820 FOR j=1 TO 13
830 car=VAL(LEFT$(posic$(i,j-1),3)):carp=VAL(LEFT$(posic$(i,j),3))
840 IF (car=13) AND (carp=1) THEN conta=conta+1
845 IF (car=1) AND (carp=1) THEN conta=conta+1
850 NEXT j,i:IF conta=4 THEN 870
860 conta=0:GOTO 340
865 REM ----- ES EL FINAL -----
870 SOUND 1,10,15
875 CLS:PRINT "HA TERMINADO EL JUEGO.VAMOS A VER COMO ESTAN LAS CARTAS"
877 REM ----- CALCULA CARTAS COLOCADAS CORRECTAMENTE -----
880 FOR i=1 TO 4
890 FOR j=1 TO 13
900 carta=VAL(LEFT$(posic$(i,j),3))
910 IF carta=j+1 THEN contador=contador+1
920 NEXT j,i
930 FOR i=1 TO 10000:NEXT:CLS
940 LOCATE 20,12:PRINT "HA COLOCADO CORRECTAMENTE ";contador+4;" CARTAS"
950 FOR i=1 TO 15000:NEXT
955 REM ----- PARA VOLVER A JUGAR -----
960 INPUT "Quieres volver a jugar ";y$:y%=LOWER$(y$)
970 IF LEFT$(y$,1)="s" THEN RUN
980 END

```